// File: /contracts/ForeverMarketV2.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/utils/cryptography/MerkleProof.sol";
import "./ForeverSplitter.sol";

interface IWETH9 {
  function approve(address guy, uint wad) external;
  function deposit() external payable;
  function withdraw(uint wad) external;
  function transferFrom(address src, address dst, uint wad) external;
}
interface ICryptoPunk {
  function balanceOf(address owner) external returns (uint256 count);
  function punkIndexToAddress(uint256 punkIndex) external returns (address);
  function punksOfferedForSale(uint256 punkIndex) external returns (bool, uint256, address, uint256, address);
  function buyPunk(uint punkIndex) external payable;
  function transferPunk(address to, uint punkIndex) external;
  function withdraw() external;
}

error InvalidBid();
error InvalidFilter();
error InvalidSignature();
error SignatureExpired();
error Unauthorized();
error WrongCommission();

contract ForeverMarketV2 is Ownable, ForeverSplitter {
  using MerkleProof for bytes32[];

  IWETH9 public immutable _wethContract;
  ICryptoPunk public immutable _punkContract;

  uint constant maxCommission = 300; // 3% in basis points
  uint public collectionCommission = 100; // 1% in basis points
  uint public traitCommission = 100; // 1% in basis points
  uint constant chainId = 1;

  enum BidStatus{ ONGOING, CANCELED, EXECUTED }
  struct Signature {
    uint8 v;
    bytes32 r;
    bytes32 s;
  }
  struct Bid {
    uint256 cancelNonce;
    uint256 bidNonce;
    uint256 deadline;
    uint256 amount;
    address bidder;
    bool collectionWide;
    bytes32 merkleTree;
    bytes32[] merkleProof;
  }

  mapping(address => uint) public cancelNonces;
  mapping(address => mapping (uint => BidStatus)) public bids;

  event BidAccepted(address indexed bidder, address indexed accepter, uint punkId, uint amount, uint8 v, bytes32 r, bytes32 s);
  event BidCanceled(address indexed bidder, uint bidNonce);
  event AllBidsCanceled(address indexed bidder, uint cancelNonce);

  constructor(address weth9Address, address cryptoPunkAddress, address foreverNftAddress) ForeverSplitter(foreverNftAddress, weth9Address) {
    _wethContract = IWETH9(weth9Address);
    _punkContract = ICryptoPunk(cryptoPunkAddress);
  }

  /**
   * @dev Set new commission for collection bids in basis points
   */
  function setCollectionCommission(uint newCollectionCommission) onlyOwner external {
    if (newCollectionCommission > maxCommission) revert WrongCommission();

    collectionCommission = newCollectionCommission;
  }

  /**
   * @dev Set new commission for collection bids in basis points
   */
  function setTraitCommission(uint newTraitCommission) onlyOwner external {
    if (newTraitCommission > maxCommission) revert WrongCommission();

    traitCommission = newTraitCommission;
  }

  /**
   * @dev Cancel a bid after authorizing it off-chain. Can only be called by the bid creator
   */
  function cancelBid(
    Bid memory bid
  ) external {
    if (block.timestamp > bid.deadline) revert SignatureExpired();
    if (msg.sender != bid.bidder) revert Unauthorized();
    if (bids[msg.sender][bid.bidNonce] != BidStatus.ONGOING) revert InvalidBid();

    bids[bid.bidder][bid.bidNonce] = BidStatus.CANCELED;
    emit BidCanceled(bid.bidder, bid.bidNonce);
  }

  /**
   * @dev Cancel all off-chain bids of the msg.sender. No deadlines or signatures are checked.
   */
  function cancelAllBids() external {
    emit AllBidsCanceled(msg.sender, cancelNonces[msg.sender]);
    cancelNonces[msg.sender]++;
  }

  function checkSignature(Signature memory sig, bytes32 hash, address bidder) pure internal returns (address) {
    address signer = ecrecover(hash, sig.v, sig.r, sig.s);
    if (signer != bidder || signer == address(0)) revert InvalidSignature();

    return signer;
  }

  function buildIntermediateSignatureHashes(
    uint cancelNonce, uint bidNonce, address bidder, uint amount,
    bool collectionWide, bytes32 merkle, uint256 deadline
  ) view internal returns (bytes32 domainHash, bytes32 structHash) {
    bytes32 eip712DomainHash = keccak256(
      abi.encode(
        keccak256("EIP712Domain(string name,string version,uint256 chainId,address verifyingContract)"),
        keccak256(bytes("ForeverMarket")),
        keccak256(bytes("1")),
        chainId,
        address(this)
      )
    );
    bytes32 hashStruct = keccak256(
      abi.encode(
        keccak256("Bid(uint cancelNonce,uint bidNonce,address bidder,uint amount,bool collectionWide,bytes32 merkleTree,uint deadline)"),
        cancelNonce,
        bidNonce,
        bidder,
        amount,
        collectionWide,
        merkle,
        deadline
      )
    );

    return (eip712DomainHash, hashStruct);
  }

  function stringToUint(string memory s) public pure returns (uint) {
    bytes memory b = bytes(s);
    uint result = 0;
    for (uint256 i = 0; i < b.length; i++) {
      uint256 c = uint256(uint8(b[i]));
      if (c >= 48 && c <= 57) {
        result = result * 10 + (c - 48);
      }
    }
    return result;
  }

  // Need this empty function declaration to accept the ETH when unwrapping WETH
  // see https://ethereum.stackexchange.com/questions/83929/while-testing-wrap-unwrap-of-eth-to-weth-on-kovan-however-the-wrap-function-i
  fallback() external payable {}
  receive() external payable {}

  function acceptBid(
    Signature calldata sig,
    Bid calldata bid,
    string calldata punkIdToSell
  ) external payable {
    if (block.timestamp > bid.deadline) revert SignatureExpired();
    if (bid.cancelNonce > cancelNonces[bid.bidder]) revert InvalidBid();

    (bytes32 eip712DomainHash, bytes32 hashStruct) = buildIntermediateSignatureHashes(
      bid.cancelNonce, bid.bidNonce, bid.bidder, bid.amount, bid.collectionWide, bid.merkleTree, bid.deadline
    );
    bytes32 hash = keccak256(abi.encodePacked("\x19\x01", eip712DomainHash, hashStruct));
    if (bids[bid.bidder][bid.bidNonce] != BidStatus.ONGOING) revert InvalidBid();

    checkSignature(sig, hash, bid.bidder);
    uint256 commission = collectionCommission;
    if (!bid.collectionWide) {
      if (!bid.merkleProof.verify(bid.merkleTree, keccak256(bytes(punkIdToSell)))) revert InvalidFilter();
      commission = traitCommission;
    }

    // set bid executed before sending punk and ETH to prevent re-entrancy attacks
    bids[bid.bidder][bid.bidNonce] = BidStatus.EXECUTED;

    // transfer WETH from bidder to SC 
    _wethContract.transferFrom(bid.bidder, address(this), bid.amount);

    // unwrap WETH (without commission) to ETH
    _wethContract.withdraw((10000 - commission) * bid.amount / 10000);

    // optional commission in WETH
    if (commission > 0) {
      uint256 commissionAmount = commission * bid.amount / 10000;
      receivePayment(address(this), commissionAmount);
    }

    uint punkId = stringToUint(punkIdToSell);
    // SC can now buy the punk
    _punkContract.buyPunk{value: (10000 - commission) * bid.amount / 10000}(punkId);

    // SC can transfer the punk to the bidder
    _punkContract.transferPunk(bid.bidder, punkId);

    emit BidAccepted(bid.bidder, msg.sender, punkId, bid.amount, sig.v, sig.r, sig.s);
  }
}

// File: /contracts/ForeverSplitter.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "@openzeppelin/contracts/utils/Context.sol";
import "@openzeppelin/contracts/token/ERC20/ERC20.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/access/Ownable.sol";

interface IERC721Snapshot {
  function createSnapshot() external returns (uint256 count);
  function balanceOfAt(address account, uint256 snapshotId) external view returns (uint256 balance);
  function totalSupplyAt(uint256 snapshotId) external view returns (uint256 supply);
  function totalSupply() external view returns (uint256 supply);
}

error InvalidAmount();
error InvalidToken();
error NoPaymentPending();
error NoRevenue();
error OutOfBounds();

/**
 * @title ForeverSplitter
 * @dev This contract allows to split ERC20 payments among a group of accounts.
 * The sender does not need to be aware that the payments will be split in this
 * way, since it is handled transparently by the contract.
 *
 * The payments are received and released in the form of an {ERC20} token. While
 * the split is done proportional to the percentage holdings of an {ForeverNft}
 * token. Whenever the contract receives (or pays) {ERC20} tokens, it
 * increaments the sanpshot id in the {ForeverNft} contract, and saves the new
 * snapshot id along with the `amount` of tokens received (or paid) and the
 * `from` address (or `to` address when paying). Any {ForeverNft} token holder
 * can get his share of payment proportional to his holdings of {ForeverNft}
 * token at the time payment was received by this contract. 
 *
 * {ForeverSplitter} follows a _pull payment_ model. This means that
 * payments are not automatically forwarded to the accounts but kept in this
 * contract, and the actual transfer is triggered as a separate step by calling
 * the {releasePayment} function.
 *
 * Note In order to receive or release payments, {ForeverSplitter} needs to
 * increment the snapshot id in the {ForeverNft} token. For this
 * {ForeverSplitter} needs {SNAPSHOT_CREATOR} role in {ForeverNft}. So,
 * after deploying it, grant it {SNAPSHOT_CREATOR} role in {ForeverNft} token.
 *
 * Note In case, this contract receives any {ERC20} tokens other than the token
 * set for payments (set at build time), the owner of the contract can call
 * {rescueTokens} function to transfer them to some address.
 */
contract ForeverSplitter is Context, Ownable {
  struct Received {
    uint256 snapshotId;
    uint256 amount;
    address from;
  }
  struct Payment {
    uint256 snapshotId;
    uint256 amount;
    address to;
  }

  Received[] private _received;
  mapping(address => uint256) private _totalReceivedFrom;
  uint256 private _totalReceived;

  mapping(address => Payment[]) private _payments;
  mapping(address => uint256) private _totalPaidTo;
  uint256 private _totalPaid;

  IERC721Snapshot immutable private _sharesToken;
  IERC20 immutable private _paymentToken;

  /**
   * @dev Emitted when a payment is received.
   * @param from Address from which payment is received.
   * @param amount Amount of `_paymentToken` received.
   */
  event PaymentReceived(address from, uint256 amount);
  /**
   * @dev Emitted when a payment is released.
   * @param to Address to which payment is released.
   * @param amount Amount of `_paymentToken` released.
   */
  event PaymentReleased(address to, uint256 amount);

  /**
   * @dev Set the values for {_sharesToken} and {_paymentToken}.
   * @param sharesToken_ Address of the {ForeverNft} token to use.
   * @param paymentTokenAddress Address of the {ERC20} token to use.
   *
   * Note All two of these values are immutable: they can only be set once
   * during construction.
   */
  constructor(address sharesToken_, address paymentTokenAddress) {
    _sharesToken = IERC721Snapshot(sharesToken_);
    _paymentToken = IERC20(paymentTokenAddress);
  }

  /**
   * @dev Get the total amount of `_paymentToken` released so far.
   */
  function totalPaid() public view returns (uint256) {
    return _totalPaid;
  }

  /**
   * @dev Get the total amount of `_paymentToken` released to `payee` so far.
   * @param payee Address for which total paid amount is desired.
   */
  function totalPaidTo(address payee) public view returns (uint256) {
    return _totalPaidTo[payee];
  }

  /**
   * @dev Get the total number of payments released to `payee` so far.
   */
  function releasedPaymentsCount(address payee) public view returns (uint256) {
    return _payments[payee].length;
  }

  /**
   * @dev Get the `pos`-th released payment to `payee`.
   * @param payee Address for which payment data is desired.
   * @param pos Index of the released payment.
   *
   * Requirements:
   *
   * - `pos` must be a valid index.
   */
  function releasedPaymentsData(address payee, uint256 pos) public view returns (Payment memory) {
    if (pos >= releasedPaymentsCount(payee)) revert OutOfBounds();
    return _payments[payee][pos];
  }

  /**
   * @dev Get the total amount of `_paymentToken` received so far.
   */
  function totalReceived() public view returns (uint256) {
    return _totalReceived;
  }

  /**
   * @dev Get the total amount of `_paymentToken` received from `payer` so far.
   * @param payer Address for which total received amount is desired.
   */
  function totalReceivedFrom(address payer) public view returns (uint256) {
    return _totalReceivedFrom[payer];
  }

  /**
   * @dev Returns the number of payments received so far.
   */
  function receivedPaymentsCount() public view returns (uint256) {
    return _received.length;
  }

  /**
   * @dev Get the `pos`-th received payment.
   * @param pos Index of the received payment.
   *
   * Requirements:
   *
   * - `pos` must be a valid index.
   */
  function receivedPaymentsData(uint256 pos) public view returns (Received memory) {
    if (pos >= receivedPaymentsCount()) revert OutOfBounds();
    return _received[pos];
  }

  /**
   * @dev Get the pending payment for `payee`.
   * @param payee Address for which pending payment is desired.
   */
  function paymentPending(address payee) public view returns(uint256 currentPayment) {
    Payment[] storage payments = _payments[payee];
    uint256 lastPaymentSnapshot = payments.length == 0 ? 0 : payments[payments.length - 1].snapshotId;

    for (uint256 i = _received.length; i > 0; --i) {
      uint256 receiveSnapshot = _received[i - 1].snapshotId;
      if (lastPaymentSnapshot > receiveSnapshot) {
        break;
      }
      uint256 sharesInReceiveSnapshot = _sharesToken.balanceOfAt(payee, receiveSnapshot);
      uint256 totalSharesInReceiveSnapshot = _sharesToken.totalSupplyAt(receiveSnapshot);
      currentPayment += (_received[i - 1].amount * sharesInReceiveSnapshot) / totalSharesInReceiveSnapshot;
    }
  }

  /**
   * @dev Receives `_paymentToken` from `sender`.
   * @param sender Address of the sender.
   * @param amount Amount of tokens to receive.
   *
   * Emits a {PaymentReceived} event.
   * Emits a {Snapshot} event on {ForeverNft} contract.
   *
   * Requirements:
   *
   * - Total shares must be non-zero at the time of receiving payment. Else,
   *   payment will get locked in this contract forever with no one to claim
   *   it.
   * - `amount` must be non-zero.
   * - Sender must have already allowed {ForeverSplitter} to draw
   *   `amount` tokens from `sender` address.
   * - This contract must have {SNAPSHOT_CREATOR} role in {ForeverNft}.
   */
  function receivePayment(address sender, uint256 amount) internal {
    if (_sharesToken.totalSupply() <= 0) revert NoRevenue();
    if (amount <= 0) revert InvalidAmount();

    _paymentToken.transferFrom(sender, address(this), amount);
    emit PaymentReceived(sender, amount);

    _totalReceived += amount;
    _totalReceivedFrom[sender] += amount;
    uint256 nextSnapshotId = _sharesToken.createSnapshot();
    _received.push(Received({
      snapshotId: nextSnapshotId,
      amount: amount,
      from: sender
    }));
  }

  /**
   * @dev Releases payment (if any) to the sender of the call.
   *
   * Emits a {PaymentReleased} event.
   * Emits a {Snapshot} event on {ForeverNft} contract.
   *
   * Requirements:
   *
   * - Sender must have non-zero pending payment.
   * - This contract must have {SNAPSHOT_CREATOR} role in {ForeverNft}.
   */
  function releasePayment() external virtual {
    address payee = _msgSender();
    uint256 payment = paymentPending(payee);
    if (payment <= 0) revert NoPaymentPending();

    _paymentToken.transfer(payee, payment);
    emit PaymentReleased(payee, payment);

    _totalPaid += payment;
    _totalPaidTo[payee] += payment;
    uint256 nextSnapshotId = _sharesToken.createSnapshot();
    _payments[payee].push(Payment({
      snapshotId: nextSnapshotId,
      amount: payment,
      to: payee
    }));
  }

  /**
   * @dev Sends {ERC20} tokens other than the `_paymentToken` to a given address.
   * @param token Address of the ERC20 token.
   * @param to Address to which the tokens are to be transferred.
   * @param amount Amount of tokens to transfer.
   *
   * Requirements:
   *
   * - Only the owner of the contract can call this.
   * - `token` must not be the same as `_paymentToken`.
   */
  function rescueTokens(address token, address to, uint256 amount) external onlyOwner {
    if (token == address(_paymentToken)) revert InvalidToken();

    IERC20(token).transfer(to, amount);
  }
}

// File: @openzeppelin/contracts/utils/cryptography/MerkleProof.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/cryptography/MerkleProof.sol)

pragma solidity ^0.8.0;

/**
 * @dev These functions deal with verification of Merkle Tree proofs.
 *
 * The tree and the proofs can be generated using our
 * https://github.com/OpenZeppelin/merkle-tree[JavaScript library].
 * You will find a quickstart guide in the readme.
 *
 * WARNING: You should avoid using leaf values that are 64 bytes long prior to
 * hashing, or use a hash function other than keccak256 for hashing leaves.
 * This is because the concatenation of a sorted pair of internal nodes in
 * the merkle tree could be reinterpreted as a leaf value.
 * OpenZeppelin's JavaScript library generates merkle trees that are safe
 * against this attack out of the box.
 */
library MerkleProof {
    /**
     * @dev Returns true if a `leaf` can be proved to be a part of a Merkle tree
     * defined by `root`. For this, a `proof` must be provided, containing
     * sibling hashes on the branch from the leaf to the root of the tree. Each
     * pair of leaves and each pair of pre-images are assumed to be sorted.
     */
    function verify(
        bytes32[] memory proof,
        bytes32 root,
        bytes32 leaf
    ) internal pure returns (bool) {
        return processProof(proof, leaf) == root;
    }

    /**
     * @dev Calldata version of {verify}
     *
     * _Available since v4.7._
     */
    function verifyCalldata(
        bytes32[] calldata proof,
        bytes32 root,
        bytes32 leaf
    ) internal pure returns (bool) {
        return processProofCalldata(proof, leaf) == root;
    }

    /**
     * @dev Returns the rebuilt hash obtained by traversing a Merkle tree up
     * from `leaf` using `proof`. A `proof` is valid if and only if the rebuilt
     * hash matches the root of the tree. When processing the proof, the pairs
     * of leafs & pre-images are assumed to be sorted.
     *
     * _Available since v4.4._
     */
    function processProof(bytes32[] memory proof, bytes32 leaf) internal pure returns (bytes32) {
        bytes32 computedHash = leaf;
        for (uint256 i = 0; i < proof.length; i++) {
            computedHash = _hashPair(computedHash, proof[i]);
        }
        return computedHash;
    }

    /**
     * @dev Calldata version of {processProof}
     *
     * _Available since v4.7._
     */
    function processProofCalldata(bytes32[] calldata proof, bytes32 leaf) internal pure returns (bytes32) {
        bytes32 computedHash = leaf;
        for (uint256 i = 0; i < proof.length; i++) {
            computedHash = _hashPair(computedHash, proof[i]);
        }
        return computedHash;
    }

    /**
     * @dev Returns true if the `leaves` can be simultaneously proven to be a part of a merkle tree defined by
     * `root`, according to `proof` and `proofFlags` as described in {processMultiProof}.
     *
     * CAUTION: Not all merkle trees admit multiproofs. See {processMultiProof} for details.
     *
     * _Available since v4.7._
     */
    function multiProofVerify(
        bytes32[] memory proof,
        bool[] memory proofFlags,
        bytes32 root,
        bytes32[] memory leaves
    ) internal pure returns (bool) {
        return processMultiProof(proof, proofFlags, leaves) == root;
    }

    /**
     * @dev Calldata version of {multiProofVerify}
     *
     * CAUTION: Not all merkle trees admit multiproofs. See {processMultiProof} for details.
     *
     * _Available since v4.7._
     */
    function multiProofVerifyCalldata(
        bytes32[] calldata proof,
        bool[] calldata proofFlags,
        bytes32 root,
        bytes32[] memory leaves
    ) internal pure returns (bool) {
        return processMultiProofCalldata(proof, proofFlags, leaves) == root;
    }

    /**
     * @dev Returns the root of a tree reconstructed from `leaves` and sibling nodes in `proof`. The reconstruction
     * proceeds by incrementally reconstructing all inner nodes by combining a leaf/inner node with either another
     * leaf/inner node or a proof sibling node, depending on whether each `proofFlags` item is true or false
     * respectively.
     *
     * CAUTION: Not all merkle trees admit multiproofs. To use multiproofs, it is sufficient to ensure that: 1) the tree
     * is complete (but not necessarily perfect), 2) the leaves to be proven are in the opposite order they are in the
     * tree (i.e., as seen from right to left starting at the deepest layer and continuing at the next layer).
     *
     * _Available since v4.7._
     */
    function processMultiProof(
        bytes32[] memory proof,
        bool[] memory proofFlags,
        bytes32[] memory leaves
    ) internal pure returns (bytes32 merkleRoot) {
        // This function rebuild the root hash by traversing the tree up from the leaves. The root is rebuilt by
        // consuming and producing values on a queue. The queue starts with the `leaves` array, then goes onto the
        // `hashes` array. At the end of the process, the last hash in the `hashes` array should contain the root of
        // the merkle tree.
        uint256 leavesLen = leaves.length;
        uint256 totalHashes = proofFlags.length;

        // Check proof validity.
        require(leavesLen + proof.length - 1 == totalHashes, "MerkleProof: invalid multiproof");

        // The xxxPos values are "pointers" to the next value to consume in each array. All accesses are done using
        // `xxx[xxxPos++]`, which return the current value and increment the pointer, thus mimicking a queue's "pop".
        bytes32[] memory hashes = new bytes32[](totalHashes);
        uint256 leafPos = 0;
        uint256 hashPos = 0;
        uint256 proofPos = 0;
        // At each step, we compute the next hash using two values:
        // - a value from the "main queue". If not all leaves have been consumed, we get the next leaf, otherwise we
        //   get the next hash.
        // - depending on the flag, either another value for the "main queue" (merging branches) or an element from the
        //   `proof` array.
        for (uint256 i = 0; i < totalHashes; i++) {
            bytes32 a = leafPos < leavesLen ? leaves[leafPos++] : hashes[hashPos++];
            bytes32 b = proofFlags[i] ? leafPos < leavesLen ? leaves[leafPos++] : hashes[hashPos++] : proof[proofPos++];
            hashes[i] = _hashPair(a, b);
        }

        if (totalHashes > 0) {
            return hashes[totalHashes - 1];
        } else if (leavesLen > 0) {
            return leaves[0];
        } else {
            return proof[0];
        }
    }

    /**
     * @dev Calldata version of {processMultiProof}.
     *
     * CAUTION: Not all merkle trees admit multiproofs. See {processMultiProof} for details.
     *
     * _Available since v4.7._
     */
    function processMultiProofCalldata(
        bytes32[] calldata proof,
        bool[] calldata proofFlags,
        bytes32[] memory leaves
    ) internal pure returns (bytes32 merkleRoot) {
        // This function rebuild the root hash by traversing the tree up from the leaves. The root is rebuilt by
        // consuming and producing values on a queue. The queue starts with the `leaves` array, then goes onto the
        // `hashes` array. At the end of the process, the last hash in the `hashes` array should contain the root of
        // the merkle tree.
        uint256 leavesLen = leaves.length;
        uint256 totalHashes = proofFlags.length;

        // Check proof validity.
        require(leavesLen + proof.length - 1 == totalHashes, "MerkleProof: invalid multiproof");

        // The xxxPos values are "pointers" to the next value to consume in each array. All accesses are done using
        // `xxx[xxxPos++]`, which return the current value and increment the pointer, thus mimicking a queue's "pop".
        bytes32[] memory hashes = new bytes32[](totalHashes);
        uint256 leafPos = 0;
        uint256 hashPos = 0;
        uint256 proofPos = 0;
        // At each step, we compute the next hash using two values:
        // - a value from the "main queue". If not all leaves have been consumed, we get the next leaf, otherwise we
        //   get the next hash.
        // - depending on the flag, either another value for the "main queue" (merging branches) or an element from the
        //   `proof` array.
        for (uint256 i = 0; i < totalHashes; i++) {
            bytes32 a = leafPos < leavesLen ? leaves[leafPos++] : hashes[hashPos++];
            bytes32 b = proofFlags[i] ? leafPos < leavesLen ? leaves[leafPos++] : hashes[hashPos++] : proof[proofPos++];
            hashes[i] = _hashPair(a, b);
        }

        if (totalHashes > 0) {
            return hashes[totalHashes - 1];
        } else if (leavesLen > 0) {
            return leaves[0];
        } else {
            return proof[0];
        }
    }

    function _hashPair(bytes32 a, bytes32 b) private pure returns (bytes32) {
        return a < b ? _efficientHash(a, b) : _efficientHash(b, a);
    }

    function _efficientHash(bytes32 a, bytes32 b) private pure returns (bytes32 value) {
        /// @solidity memory-safe-assembly
        assembly {
            mstore(0x00, a)
            mstore(0x20, b)
            value := keccak256(0x00, 0x40)
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.openzeppelin.com/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
            // Overflow not possible: the sum of all balances is capped by totalSupply, and the sum is preserved by
            // decrementing then incrementing.
            _balances[to] += amount;
        }

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        unchecked {
            // Overflow not possible: balance + amount is at most totalSupply + amount, which is checked above.
            _balances[account] += amount;
        }
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
            // Overflow not possible: amount <= accountBalance <= totalSupply.
            _totalSupply -= amount;
        }

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}